

Solution

SP4R05s03.sas

The **AmesHousing** data set was used to complete this exercise.

Using the SYMPUTX Subroutine

a. The SYMPUTX subroutine enables you to create a macro variable inside a DATA step. Navigate to the online documentation for a complete description. Open **SP4R05e03.sas.** Submit the code (shown below) and analyze both the code and log output. What does this code do?

```
data _NULL_;
   x=-3;
   df=5;
   p=(1-probt(abs(x),df))*2;
   call symputx('sig_level',p);
run;

%put The significance level for the two-tailed t test is &sig_level;
```

```
The significance level for the two-tailed t test is 0.0300992479
```

This code uses a DATA _NULL_ step to create a macro variable for the significance level of a two-sided *t* test with five degrees of freedom and a test value of -3.

Selected functions and subroutines:

PROBT(x,df) returns the probability that an observation form a Student's ***t*** distribution, with degrees of freedom **df**, is less than or equal to **x**.

SYMPUTX assigns a value to a macro variable and removes both leading and trailing blanks.

b. An alternative method to creating the macro variable in Exercise 2 is to use the SYMPUTX subroutine. Use a DATA _NULL_ step, a SET statement, and the SYMPUTX routine to create a macro variable for the median of the **saleprice** variable. Use the %PUT statement to ensure that the macro variable is created correctly.

```
proc means data=sp4r.ameshousing;
   var saleprice;
   output out=stats median=sp_med;
run;

data _null_;
   set stats;
   call symputx('med',sp_med);
run;

%put The median of the Sale Price variable is &med;
```

```
The median of the Sale Price variable is 135000
```