



Solution

SP4R04s01.sas

1. Using the DO loop and Creating Random Data Sets

- a. Navigate to the **SAS RAND function** page and choose a few functions to practice generating random numbers. Create a data table with at least two variables of random numbers and at least 10 observations. Be sure to use a random seed of your choice.

```
data sp4r.random;
  call streaminit(123);
  do i=1 to 10;
    rt = rand('T',5);
    rf = rand('F',3,4);
    ru = int(rand('Uniform')*10);
    output;
  end;
run;

proc print data=sp4r.random;
run;
```

Obs	i	rt	rf	ru
1	1	0.15554	0.57611	3
2	2	-0.71020	0.15053	2
3	3	-0.02583	0.04516	9
4	4	0.73364	0.25261	7
5	5	0.18336	0.88293	4
6	6	0.13730	1.50425	9
7	7	0.90893	2.18254	9
8	8	0.04611	0.10342	8
9	9	2.41523	0.55436	5
10	10	0.20044	1.59396	1

- b. Create a new data table with the same random variables that you specified from the previous step. Create a variable called **Class** that groups the first five observations into class 1 and the second five into class 2. Drop the nested DO loop index variable from the data table and add a sequence from 1 to 10. Print the data upon completion.

```
data sp4r.random (drop=j);
  call streaminit(123);
  do class=1 to 2;
    do j=1 to 5;
      sequence + 1;
      rt = rand('T',5);
      rf = rand('F',3,4);
      ru = int(rand('Uniform')*10);
      output;
    end;
  end;
run;

proc print data=sp4r.random;
run;
```

	Obs	class	sequence	rt	rf	ru
	1	1	1	0.15554	0.57611	3
	2	1	2	-0.71020	0.15053	2
	3	1	3	-0.02583	0.04516	9
	4	1	4	0.73364	0.25261	7
	5	1	5	0.18336	0.88293	4
	6	2	6	0.13730	1.50425	9
	7	2	7	0.90893	2.18254	9
	8	2	8	0.04611	0.10342	8
	9	2	9	2.41523	0.55436	5
	10	2	10	0.20044	1.59396	1

c. Run the SAS code below. What do you notice?

```
data test;
  do i=1 to 2;
    output;
  end;
run;

proc print data=test;
run;

data test;
  set test;
  do j=1 to 5;
    output;
  end;
run;

proc print data=test;
run;
```

	Obs	i
	1	1
	2	2

	Obs	i	j
	1	1	1
	2	1	2
	3	1	3
	4	1	4
	5	1	5
	6	2	1
	7	2	2
	8	2	3
	9	2	4
	10	2	5

The loop iterates through each observation in the data table.